Newsgroup: it.comp.lang.visual-basic
Titolo: aprire un documento

Scusate la mia ignoranza ma volevo sapere quale e' il comando da inserire
per far si che si apra un documento di word con un pulsante.Grazie,Ciao

 quale e' il comando da inserire per far si che si apra un documento di
word con un pulsante.
Consulta la guida alla voce Shell. Poi visita anche il Sito comune di questo
ng:
 
http://www.murialdo.it/it_lang_vb
--
Ciao.
Maurizio
--------
?SPQR(C)
 X
--------

     
http://www.murialdo.it/it_lang_vb
Mi chiedi un esempio perche' non hai la guida e nelle faq non hai trovato
niente. Male, molto male! Cmq ecco l'esempio. Studialo a fondo prima di
chiedere spiegazioni. Aspetto una tua risposta sul ng in cui scrivi:
    Grazie (facoltativo). Capito tutto. Ho usato il metodo:
    [1]    [2]    [3]    [4]    (Crocettina, prego)
    Perche': _____________________________________________
(Io la crocettina la metto su [4]).
Option Explicit
Private Enum myShowCmd
  SW_HIDE
  SW_SHOWNORMAL
  SW_SHOWMINIMIZED
  SW_SHOWMAXIMIZED
  SW_SHOWNOACTIVATE
  SW_SHOW
  SW_MINIMIZE
  SW_SHOWMINNOACTIVE
  SW_SHOWNA
  SW_RESTORE
End Enum
' HINSTANCE ShellExecute(
'   HWND    hWnd,         // handle to parent window
'   LPCTSTR lpOperation,  // pointer to string that specifies
'                         //    operation to perform
'   LPCTSTR lpFile,       // pointer to filename or folder
'                         //    name string
'   LPCTSTR lpParameters, // pointer to string that specifies
'                         //    executable-file parameters
'   LPCTSTR lpDirectory,  // pointer to string that specifies
'                         //  default directory
'   INT     nShowCmd      // whether file is shown when opened
' );
Private Declare Function ShellExecute _
  Lib "shell32.dll" Alias "ShellExecuteA" ( _
    ByVal hWnd As Long _
  , ByVal lpOperation As String _
  , ByVal lpFile As String _
  , ByVal lpParameters As String _
  , ByVal lpDirectory As String _
  , ByVal nShowCmd As myShowCmd _
  ) As Long
Private Sub Command1_Click( _
)
On Error GoTo Command1_Click_Err
  Const c_wdAppPathName As String _
    = "c:\programmi\microsoft office\office\winword.exe"
  Const c_wdClass       As String _
    = "Word.Application"
  Const c_wdExeName     As String _
    = "WinWord"
  Dim strPathName As String
  Screen.MousePointer = vbHourglass
  Command1.Enabled = False
  DoEvents
  strPathName = Text1.Text
  If Len(strPathName) Then
    strPathName _
      = strPathName _
      & IIf(Right$(strPathName, 1) = "\", "", "\") _
      & Text2.Text
  Else
    strPathName = Text2.Text
  End If
  ' *1*
  '
  If Len(strPathName) Then
    strPathName _
      = c_wdAppPathName & Chr$(32) _
      & Chr$(34) & strPathName & Chr$(34)
  Else
    strPathName _
      = c_wdAppPathName
  End If
  Shell strPathName, vbNormalFocus
  ' *2*
  '
  'If Len(strPathName) Then
  '  ShellExecute _
  '      hWnd _
  '    , "open" _
  '    , strPathName _
  '    , vbNullString _
  '    , vbNullString _
  '    , SW_SHOWNORMAL
  '
  'Else
  '  ShellExecute _
  '      hWnd _
  '    , vbNullString _
  '    , c_wdExeName _
  '    , vbNullString _
  '    , vbNullString _
  '    , SW_SHOWNORMAL
  '
  'End If
  ' *3*
  '
  'If Len(strPathName) Then
  '  strPathName _
  '    = "start" & Chr$(32) _
  '    & Chr$(34) & strPathName & Chr$(34)
  '
  'Else
  '  strPathName _
  '    = "start" & Chr$(32) _
  '    & c_wdExeName
  '
  'End If
  'Shell strPathName, vbHide
  ' *4*
  '
  'If Len(strPathName) Then
  '  GetObject(strPathName).Application.Visible = True
  '
  'Else
  '  CreateObject(c_wdClass).Visible = True
  '
  'End If
  Command1.Enabled = True
  Screen.MousePointer = vbDefault
  Exit Sub
Command1_Click_Err:
  Beep
End Sub
Private Sub Form_Load( _
)
  Label1.Caption = "&Percorso:"
  Label2.Caption = "&Documento:"
  Text1.Text = ""
  Text2.Text = ""
End Sub
